```
ACC := 40;
t := series(q^(-5) * mul( ( (1-q^n)/(1-q^(11*n)) )^12, n = 1 ..ACC), q = 0, ACC);
f := series(diff(t, q) * mul( (1-q^n)^(-2) * (1-q^(11*n))^(-2), n = 1 ..ACC), q = 0, ACC);
h := series(q*t*mul(1-q^(11*k), k = 1 ..ACC) * add(combinat[numbpart](11*k + 6)*q^k, k
    = 0 ..ACC), q = 0, ACC);
```

$$ACC := 40$$

$$t := q^{-5} - 12\,q^{-4} + 54\,q^{-3} - 88\,q^{-2} - 99\,q^{-1} + 540 - 418\,q - 648\,q^{2} + 594\,q^{3} + 836\,q^{4} + 1056\,q^{5} - 4092\,q^{6} - 353\,q^{7} + 4752\,q^{8} - 1650\,q^{9} + 3068\,q^{10} - 9768\,q^{12} - 8074\,q^{13} + 12144\,q^{14} + 27258\,q^{15} + 572\,q^{16} - 54504\,q^{17} - 4884\,q^{18} + 45045\,q^{19} - 22176\,q^{20} + 61656\,q^{21} - 104676\,q^{23} - 69564\,q^{24} + 78914\,q^{25} + 290664\,q^{26} - 72732\,q^{27} - 411180\,q^{28} + 8646\,q^{29} + 241812\,q^{30} - 117194\,q^{31} + 567996\,q^{32} - 842336\,q^{34} + O\left(q^{35}\right)$$

$$f := (-5)\,q^{-6} + 38\,q^{-5} - 91\,q^{-4} + 42\,q^{-3} + 21\,q^{-2} + 238\,q^{-1} - 884\,q - 759\,q^{2} + 812\,q^{3} + 7378\,q^{4} - 6252\,q^{5} - 6381\,q^{6} + 3346\,q^{7} - 11809\,q^{8} + 14540\,q^{9} + 41209\,q^{10} - 86204\,q^{12} - 64802\,q^{13} + 75467\,q^{14} + 228684\,q^{15} - 156394\,q^{16} - 105152\,q^{17} + 106148\,q^{18} - 448770\,q^{19} + 395312\,q^{20} + 712908\,q^{21} - 1403928\,q^{23} - 1199067\,q^{24} + 1634612\,q^{25} + 2782237\,q^{26} - 1770056\,q^{27} - 808215\,q^{28} + 822298\,q^{29} - 6583279\,q^{30} + 5287908\,q^{31} + 7492338\,q^{32} + O\left(q^{34}\right)$$

$$h := 11\,q^{-4} + 165\,q^{-3} + 748\,q^{-2} + 1639\,q^{-1} + 3553 + 4136\,q + 6347\,q^{2} + 3586\,q^{3} + 7414\,q^{4} - 4444\,q^{5} + 583\,q^{6} - 14157\,q^{7} - 6523\,q^{8} - 29590\,q^{9} + 17435\,q^{10} - 14641\,q^{11} + 34100\,q^{12} + 27863\,q^{13} + 43186\,q^{14} + 40216\,q^{15} + 12738\,q^{16} - 51216\,q^{17} - 85162\,q^{18} - 32142\,q^{19} - 268488\,q^{20} + 95194\,q^{21} - 102487\,q^{22} + 188386\,q^{23} + 135927\,q^{24} + 411906\,q^{25} + 184932\,q^{26} + 322366\,q^{27} - 386969\,q^{28} - 489467\,q^{29} - 36773\,q^{30} - 1503920\,q^{31} + 174262\,q^{32} - 556358\,q^{33} + 914298\,q^{34} + 443982\,q^{35} + O\left(q^{36}\right) \tag{1}$$

```
# Compute the algebraic relation p in Q[x,y] between t and f using an Ansatz:
n := 5;
m := 6;
p := add(add( c[i,j]*x^i*y^j, i = 0 ..min(m, max(m, n)-j)), j = 0 ..n);
IsZero := eval(p, {x = t, y = f}) :
SolveCoeffsZero := proc(IsZero, q, ACC)
    solve( {coeffs(convert(series(IsZero, q = 0, ACC), polynom), q)})
end:
p := eval(p, SolveCoeffsZero(IsZero, q, ACC)) :
p := sort(collect(primpart(p, y), y, factor), y);
```

$$n := 5$$

$$m := 6$$

$$p := c_{0,0} + c_{1,4}\,x\,y^4 + c_{1,2}\,x\,y^2 + c_{0,1}\,y + c_{2,1}\,x^2\,y + c_{0,3}\,y^3 + c_{5,0}\,x^5 + c_{4,0}\,x^4 + c_{2,0}\,x^2 + c_{2,2}\,x^2\,y^2 + c_{1,5}\,x\,y^5 + c_{0,4}\,y^4 + c_{3,1}\,x^3\,y + c_{1,0}\,x + c_{1,3}\,x\,y^3 + c_{4,1}\,x^4\,y + c_{6,0}\,x^6 + c_{2,3}\,x^2\,y^3 + c_{1,1}\,x\,y + c_{3,3}\,x^3\,y^3 + c_{3,0}\,x^3 + c_{0,5}\,y^5 + c_{5,1}\,x^5\,y + c_{3,2}\,x^3\,y^2 + c_{4,2}\,x^4\,y^2 + c_{2,4}\,x^2\,y^4 + c_{0,2}\,y^2$$

$$p := y^5 + 170\,x\,y^4 + 9345\,x^2\,y^3 + 167320\,x^3\,y^2 + x^4\,\left(3125\,x^2 - 7903458\,x + 5536128125\right) \quad \textbf{(2)}$$

```
# Compute a basis v of all integral functions i for which i/x^d is
# integral at infinity. Then write h as a linear combination of v:
d := 1;
read NormalBasis :
v := UpToDegree(p, x, y, d) :
H := add(c[i]*v[i], i = 1 ..nops(v)) :
IsZero := h - eval(H, {x = t, y = f} ) :
h := factor(eval(H, SolveCoeffsZero(IsZero, q, ACC)));
```

$$d := 1$$

$$h := -\frac{1}{625\,x^3\,(x+1331)\,(x-1331)}\left(11\,\left(829969\,x^5 - 33692\,x^4\,y + 2046\,x^3\,y^2 - 13\,x^2\,y^3 - x\,y^4 - 2094307204\,x^4 + 42341772\,x^3\,y + 44713614\,y^2\,x^2 + 1654433\,x\,y^3 + 14641\,y^4 + 1473717306875\,x^3\right)\right) \quad \textbf{(3)}$$

```
# Shorten h with "Rational Univariate Representation"
ShortenAlg := proc(h, f, x, y) local d, RUR;
    d := diff(f, y);
    RUR := rem(h*d, f, y)/d;
    map(factor, convert(RUR, parfrac, y))
end:
h := ShortenAlg(h, p, x, y);
```

$$h := 14641 + \frac{275\,x\,(x-1331)}{47\,x+y} - \frac{110\,(x+1331)\,(71\,x+3\,y)\,x}{1424\,x^2 + 89\,y\,x + y^2} \quad \textbf{(4)}$$

```
with(algcurves) :
genus(p, x, y);
```

$$1 \quad \textbf{(5)}$$

```
# The genus is 1. So the integral elements can have poles at
# infinity of order 0, 2, 3, 4, .... (the only gap is 1).
# Call those functions 1, z[2], z[3], ... then we should
# be able to write h as a polynomial in z[2], z[3].

# Compute the q-expansions of a basis v of some subspace of Q(x)[y]/(f).
# Adjust this basis to bring these q-expansions into Reduced Echelon Form.
# Return the adjusted basis, as well as q-expansions up to O(q^1).
qEchelonForm := proc(v, q, x, y, t, f, ACC)
    local n, m, w, i, j, M;
```

```
    n := nops(v);
    w := [seq(convert(series(eval(i, {x = t, y = f} ), q = 0, ACC), polynom), i = v)];
    m := min(map(ldegree, w, q));
    M := Matrix(n, 1-m, [seq(seq(coeff(i, q, j), j = m ..0), i = w)]);
    M := ⟨ M | LinearAlgebra:-IdentityMatrix(n) ⟩ ;
    M := LinearAlgebra:-ReducedRowEchelonForm( M );
    [seq([factor(add(M[i, 1-m + j]* v[j], j = 1 ..n)),
     add(M[i, j-m + 1]* q^j, j = m ..0)], i = 1 ..n)]
end:

v := qEchelonForm(v, q, x, y, t, f, ACC) :
for i in v do
    PoleOrder := -ldegree(i[2], q);
    z[PoleOrder] := ShortenAlg(i[1], p, x, y);
    zq[PoleOrder] := i[2] + O(q);
od;

# Let H be an Ansatz for the expression of h in terms of z[2] and z[3]:
H := c1 * z2^2 + c2 * z3 + c3 * z2 + c4;
IsZero := subs(z2 = z[2], z3 = z[3], x = t, y = f, h-H) :
'h' = eval(H, SolveCoeffsZero(IsZero, q, ACC));
```

$$PoleOrder := 5$$

$$z_5 := x + 12 - \frac{25\,x\,(x - 1331)}{47\,x + y} - \frac{5\,(x + 1331)\,(144\,x + 7\,y)\,x}{1424\,x^2 + 89\,y\,x + y^2}$$

$$zq_5 := \frac{1}{q^5} - \frac{1}{q} + \mathrm{O}(q)$$

$$PoleOrder := 4$$

$$z_4 := 12 - \frac{15\,x\,(x - 1331)}{22\,(47\,x + y)} - \frac{5\,(x + 1331)\,(28\,x + 19\,y)\,x}{22\,\left(1424\,x^2 + 89\,y\,x + y^2\right)}$$

$$zq_4 := \frac{1}{q^4} - \frac{2}{q} + \mathrm{O}(q)$$

$$PoleOrder := 3$$

$$z_3 := 12 + \frac{15\,x\,(x - 1331)}{22\,(47\,x + y)} - \frac{5\,(x + 1331)\,(16\,x + 3\,y)\,x}{22\,\left(1424\,x^2 + 89\,y\,x + y^2\right)}$$

$$zq_3 := \frac{1}{q^3} + \frac{1}{q} + \mathrm{O}(q)$$

$$PoleOrder := 2$$

$$z_2 := 12 + \frac{5\,x\,(x - 1331)}{22\,(47\,x + y)} - \frac{5\,(x + 1331)\,(42\,x + y)\,x}{22\,\left(1424\,x^2 + 89\,y\,x + y^2\right)}$$

$$zq_2 := \frac{1}{q^2} + \frac{2}{q} + \mathrm{O}(q)$$

$$PoleOrder := 0$$

$$z_0 := 1$$

$$zq_0 := 1 + \mathrm{O}(q)$$

$$H := c1\, z2^2 + c2\, z3 + c3\, z2 + c4$$

$$h = 11\, z2^2 + 704\, z2 + 121\, z3 + 3157 \qquad \textbf{(6)}$$

```
# Compute the algebraic relation between z[2] and z[3] using an Ansatz:
H := z3^2 + c1 * z3 + c2 * z2 * z3 + c3 * z2^3 + c4 * z2^2 + c5 * z2 + c6 :
IsZero := subs(z2 = z[2], z3 = z[3], x = t, y = f, H) :
sort( eval(H, SolveCoeffsZero(IsZero, q, ACC)), z3) = 0;
ifactor(j_invariant(lhs(%), z2, z3));
ifactor(j_invariant(p, x, y));
```

$$z3^2 + 6\, z2\, z3 + 25\, z3 - z2^3 - 2\, z2^2 + 45\, z2 + 168 = 0$$

$$-\frac{(2)^{12}\,(31)^3}{(11)^5}$$

$$-\frac{(2)^{12}\,(31)^3}{(11)^5} \qquad \textbf{(7)}$$